
	Reason for change:
	Note: cell Ids used in Reasoning and Summary are in reference to 8.1.4.1.7.1

In the current TTCN Implementation of the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common,

1. At Step  4, if the source Scell is same as Target Pcell ,
Scell is made as Normal Cell without considering the associated Pcell .
To have the implementation more intact , it is proposed to consider the transition of Pcell before changing the Scell  Normal Cell. Considering the Pcell transition at this step , has implications at step 17 ( described in Summary of change section )
This makes sure that Link between Pcell and its Scell is broken , Downlink Assignment Index is switched to Single Serving cell for Cell 1 as associated Scell is going to become target Pcell, and also remove source SCell from the SCell list in the source PCell.

2.At Step 13, during the reception of reconfiguration complete , presence of mac bearer routing is not considered based on the Scell existence ( At step 7-8 of the testcase doesn’t have an associated Scell )

3. At Step 18.1 , Reconfiguration of Scell to Normal cell is not really required as Scell is not becoming Target Pcell . If we take 8.1.4.1.7.1 as an example , Cell 2 is going to stay as Scell through out the scenario , and it was created as Scell , it can not be changed to Normal Cell.


	
	

	Summary of change:
	1. At Step 4 , before making Source Scell to Normal Cell , it is proposed to take care of the transition of corresponding Pcell to Normal cell ; It makes sure that the Link between Pcell and its Scell is broken , Downlink Assignment Index is switched to Single Serving cell for Cell 1 as associated Scell is going to become target Pcell . ( DAI related explanations are already given for 8.1.4.1.9.x test cases using  R5s220354 ) ; 
Please refer the Proposed implementation for this : calling f_NR_ReconfigurePCellToNormalCell before Scell  Normal Transition is done carefully to take care of only cell related transition and RB release and reset is done after Pdcp Handover Control completition

2. At Step 13, while receiving reconfiguration complete , MAC Bearer Routing has to be checked based on the presence of Scell . At Step 7 of the test case , Reconfiguration is sent to release the Scell and also to do handover from Cell 3  cell 12 . When Reconfiguration complete is received on Cell 12 it is not having any associated Scell , hence macBearerRouting cannot be expected to be present .

3.At Step 18.1 , Reconfiguration of Scell to Normal cell can be removed as Scell is not becoming Target Pcell/Scell . 
Having this function in place , wrongly makes Cell 2 as Normal cell during Step 7-8 of the test case as  there is no target Scell . More specifically Cell 2 was created as Scell ( it can not be converted into Normal SPCell )

A Prose CR will be raised on 38.523-2 cl 7.3.5.3.4 for this

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc101446252]Corrections required
0.1 [bookmark: _Toc101446253]Correction to the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common()
	Function name
	f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common

	Reason for change
	Note: cell Ids used in Reasoning and Summary are in reference to 8.1.4.1.7.1

In the current TTCN Implementation of the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common,

1. At Step  4, if the source Scell is same as Target Pcell ,
Scell is made as Normal Cell without considering the associated Pcell .
To have the implementation more intact , it is proposed to consider the transition of Pcell before changing the Scell  Normal Cell. Considering the Pcell transition at this step , has implications at step 17 ( described in Summary of change section )
This makes sure that Link between Pcell and its Scell is broken , Downlink Assignment Index is switched to Single Serving cell for Cell 1 as associated Scell is going to become target Pcell, and also remove source SCell from the SCell list in the source PCell.

2.At Step 13, during the reception of reconfiguration complete , presence of mac bearer routing is not considered based on the Scell existence ( At step 7-8 of the testcase doesn’t have an associated Scell )

3. At Step 18.1 , Reconfiguration of Scell to Normal cell is not really required as Scell is not becoming Target Pcell . If we take 8.1.4.1.7.1 as an example , Cell 2 is going to stay as Scell through out the scenario , and it was created as Scell , it can not be changed to Normal Cell.


	Summary of change
	1. At Step 4 , before making Source Scell to Normal Cell , it is proposed to take care of the transition of corresponding Pcell to Normal cell ; It makes sure that the Link between Pcell and its Scell is broken , Downlink Assignment Index is switched to Single Serving cell for Cell 1 as associated Scell is going to become target Pcell . ( DAI related explanations are already given for 8.1.4.1.9.x test cases using  R5s220354 ) ; 
Please refer the Proposed implementation for this : calling f_NR_ReconfigurePCellToNormalCell before Scell  Normal Transition is done carefully to take care of only cell related transition and RB release and reset is done after Pdcp Handover Control completition

2. At Step 13, while receiving reconfiguration complete , MAC Bearer Routing has to be checked based on the presence of Scell . At Step 7 of the test case , Reconfiguration is sent to release the Scell and also to do handover from Cell 3  cell 12 . When Reconfiguration complete is received on Cell 12 it is not having any associated Scell , hence macBearerRouting cannot be expected to be present .

3.At Step 18.1 , Reconfiguration of Scell to Normal cell can be removed as Scell is not becoming Target Pcell/Scell . 
Having this function in place , wrongly makes Cell 2 as Normal cell during Step 7-8 of the test case as  there is no target Scell . More specifically Cell 2 was created as Scell ( it can not be converted into Normal SPCell )

	TTCN module
	Common\NR5GC\NR5GC_Handover.ttcn

	MCC160 Comment
	



    
Before Change:
	function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common(NR_CellId_Type                               p_SourcePCellId,
                                                          NR_CellId_Type                               p_SourceSCellId,
                                                          NR_CellId_Type                               p_TargetPCellId,
                                                          template(omit) NR_CellId_Type                p_TargetSCellId,
                                                          template(omit) CellGroupConfig               p_SCGConfig_SCell,
                                                          RNTI_Value                                   p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                          boolean                                      p_RBConfig_KeyChange := true,
                                                          template (value) DL_DCCH_Message             p_RRCReconfiguration,
                                                          template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,
                                                          template (value) NR_DRBInfoList_Type         p_DrbInfoList,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,
                                                          template (present) UL_DCCH_Message           p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,
                                                          NextHopChainingCount                         p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                          template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit
                                                      ) runs on NR_BASE_PTC
  { //@sic R5s201249: added p_SCGConfig_SCell R5-213516: added MacBearerRouting sic@
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(p_DrbInfoList);
    var IntegerList_Type v_DrbIdList;
    var template(value) TimingInfo_Type v_TimingInfo;  //@sic R5s201387 sic@
    
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourcePCellId, 700); // 700 ms // Time for schedule of DL RRC message
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 10);  // time for Target SCell-> PCell configuration SS configuration
    v_Timing2 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 20);  // time for Target PCell SS configuration

    // 1. Source PCell: Stop periodic TA.
    // NOTE 1: Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.
    f_NR_ULGrantConfiguration_Common(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // 2. Target PCell: Configure target PCell for no RACH response transmission.
    //Calculate T in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_TargetPCellId);
    f_NR_SS_RachProcedureConfig(p_TargetPCellId, cs_NR_RachProcedureConfig_NoResponse, v_TimingInfo);

    // 3. Source PCell: Schedule the transmission of RRCReconfiguration message to UE requesting Handover to target PCell and SCell.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration, crs_MacBearerRouting(p_SourcePCellId))); //@sic R5-213516 sic@

   // At time T1 if target PCell is the same as source SCell, else at time "Now", steps 4-6:
   if (p_TargetPCellId == p_SourceSCellId) {
        // 4. Target PCell: If target PCell is same as source SCell, configure SS for target PCell to be converted from a SCell to PCell.
        // 5. Target PCell: Configuration of DRBs
        //Calculate T1 in the TargetPCell
        v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing1), p_TargetPCellId);
        f_NR_ReconfigSCellToNormalCell(p_TargetPCellId, v_RadioBearerList, v_TimingInfo);
   } else {
        // At time Now, steps 5-6:
        // 5. Target PCell: Configuration of DRBs
        f_NR_SS_CommonRadioBearerConfig(p_TargetPCellId, v_RadioBearerList); //@sic R5s201249 sic@
   }

    //6. Transfer of the PDCP Count for AM DRBs and SRBs (if applied) from source to target PCell:
    //   a) Source PCell: Get PDCP COUNT.
    //   b) Target PCell: Set PDCP COUNT.
    //   NOTE 2: No further sending/receiving of DRB data before the HO is done.
    //   NOTE 3: For AM DRBs the PDCP count is maintained, for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    if (p_TargetPCellId == p_SourceSCellId) {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange, cs_TimingInfo_NR(v_Timing1));
    } else {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s201249 sic@
    }
    
   //At time T2, steps 7-12:
    //7. Target PCell: Inform the SS about the HO and about the source PCell id.
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo_NR(v_Timing2)));
   
    //8. Target PCell: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE 4: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI
    f_NR_CellInfo_SetRNTI(p_TargetPCellId, p_RNTI_Value);
    //Calculate T2 in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), p_TargetPCellId);
    if (isvalue(p_TargetSCellId)) { //@sic R5s210079 sic@
        //@sic R5s201249 sic@
        f_NR_CellInfo_SetRNTI(valueof(p_TargetSCellId), p_RNTI_Value);
        f_NR_SS_C_RNTI_Config(valueof(p_TargetSCellId), p_RNTI_Value, cs_TimingInfo_NR(v_Timing2)); //@sic R5s210389 sic@
    }
    f_NR_SS_C_RNTI_Config(p_TargetPCellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing2), tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369, R5s210389 sic@
    f_NR_SS_RachProcedureConfigHO(p_TargetPCellId, p_RachProcedureConfig, v_TimingInfo, p_RACH_ConfigDedicated);
    
    //9. Target PCell: Activate security.
    if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s190795 R5s200159 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_TargetPCellId, p_NCC, v_TimingInfo);
    } else {
      f_NR_SS_AS_ActivateSecurity_NoRefresh(p_TargetPCellId, v_TimingInfo);  //@sic R5s200102 sic@
    }

    //10. Target PCell: Configure UL grant configuration ("OnSR", periodic TA is not started).
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, v_TimingInfo, -, -, p_UL_GrantConfig_TargetCell);

    //11. Target PCell: Configure Target PCell as PCell.
    //12. Target SCell: Configure Target SCell as SCell with new PCell association.
    //NOTE 5: The FollowOnFlag is set to true in the ASP reconfiguring PCell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
      f_NR_ConfigPCell_AddRelSCells(p_TargetPCellId, {valueof(p_TargetSCellId)}, tsc_PCellIndex_0, v_TimingInfo, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369 sic@
      //Calculate T2 in the TargetSCell
      v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), valueof(p_TargetSCellId));
      f_NR_ConfigSCell(valueof(p_TargetSCellId), p_SCGConfig_SCell, v_TimingInfo);
    }
    
   // After time T2 (without activation time):
    // 13. Target PCell: Receive RRCReconfigurationComplete.
    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete, -, crs_MacBearerRouting(p_TargetPCellId))); //@sic R5-213516 sic@
    
    //14. Target Pell: Start periodic TA.
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);
    
    //15. Target Pell: Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetPCellId));

    //16. Target PCell: Re-configure RACH procedure as for initial access.
    f_NR_SS_RachProcedureConfig_Def(p_TargetPCellId);
        
    //17. Source PCell: If source PCell is not target SCell
    //17.1 Reset SRBs and release DRBs.
    //17.2 Configure from PCell to normal cell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
        if (p_SourcePCellId != valueof(p_TargetSCellId)) {
            v_DrbIdList := f_NR_GetActiveDRBs(p_SourcePCellId);
            f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, -, v_DrbIdList );
        }
    }
    
    //18. Source SCell: If source SCell is neither target PCell nor target SCell:
    //18.1 Re-configure source SCell to normal cell and remove source SCell from the SCell list in the source PCell.
    if (not(match(p_SourceSCellId, p_TargetPCellId)) and not match(p_SourceSCellId, p_TargetSCellId)) {
        f_NR_ReconfigSCellToNormalCell(p_SourceSCellId, v_RadioBearerList, -);
        f_NR_Rel_SCell_Common( p_SourcePCellId,  p_SourceSCellId, tsc_PCellIndex_0, cs_TimingInfo_Now);//Release SCell from PCell @sic R5s201249 sic@
    }
   
  }



After Change:
	function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common(NR_CellId_Type                               p_SourcePCellId,
                                                          NR_CellId_Type                               p_SourceSCellId,
                                                          NR_CellId_Type                               p_TargetPCellId,
                                                          template(omit) NR_CellId_Type                p_TargetSCellId,
                                                          template(omit) CellGroupConfig               p_SCGConfig_SCell,
                                                          RNTI_Value                                   p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                          boolean                                      p_RBConfig_KeyChange := true,
                                                          template (value) DL_DCCH_Message             p_RRCReconfiguration,
                                                          template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,
                                                          template (value) NR_DRBInfoList_Type         p_DrbInfoList,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,
                                                          template (present) UL_DCCH_Message           p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,
                                                          NextHopChainingCount                         p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                          template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit
                                                      ) runs on NR_BASE_PTC
  { //@sic R5s201249: added p_SCGConfig_SCell R5-213516: added MacBearerRouting sic@
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(p_DrbInfoList);
    var IntegerList_Type v_DrbIdList;
    var template(value) TimingInfo_Type v_TimingInfo;  //@sic R5s201387 sic@
    var boolean ScellBecomingPcell := false; //WA#WI=944981
    
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourcePCellId, 700); // 700 ms // Time for schedule of DL RRC message
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 10);  // time for Target SCell-> PCell configuration SS configuration
    v_Timing2 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 20);  // time for Target PCell SS configuration

    // 1. Source PCell: Stop periodic TA.
    // NOTE 1: Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.
    f_NR_ULGrantConfiguration_Common(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // 2. Target PCell: Configure target PCell for no RACH response transmission.
    //Calculate T in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_TargetPCellId);
    f_NR_SS_RachProcedureConfig(p_TargetPCellId, cs_NR_RachProcedureConfig_NoResponse, v_TimingInfo);

    // 3. Source PCell: Schedule the transmission of RRCReconfiguration message to UE requesting Handover to target PCell and SCell.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration, crs_MacBearerRouting(p_SourcePCellId))); //@sic R5-213516 sic@

   // At time T1 if target PCell is the same as source SCell, else at time "Now", steps 4-6:
   if (p_TargetPCellId == p_SourceSCellId) {
        // 4. Target PCell: If target PCell is same as source SCell, configure SS for target PCell to be converted from a SCell to PCell.
        // 5. Target PCell: Configuration of DRBs
        //Calculate T1 in the TargetPCell
        //"CA HO : Source Scell is becoming Target Pcell");  //WA#WI=944981
        ScellBecomingPcell := true; //WA#WI=944981
        v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing1), p_TargetPCellId);
	      //Before making Scell to Normal cell , ensure associated Pcell - Scell Link is broken ; RB reset Source of Pcell is taken care after PDCP HO is completed
        v_DrbIdList := f_NR_GetActiveDRBs(p_SourcePCellId); //WA#WI=944981
        f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, v_TimingInfo, v_DrbIdList, ScellBecomingPcell );  //WA#WI=944981
        f_NR_ReconfigSCellToNormalCell(p_TargetPCellId, v_RadioBearerList, v_TimingInfo);
   } else {
        // At time Now, steps 5-6:
        // 5. Target PCell: Configuration of DRBs
        f_NR_SS_CommonRadioBearerConfig(p_TargetPCellId, v_RadioBearerList); //@sic R5s201249 sic@
   }

    //6. Transfer of the PDCP Count for AM DRBs and SRBs (if applied) from source to target PCell:
    //   a) Source PCell: Get PDCP COUNT.
    //   b) Target PCell: Set PDCP COUNT.
    //   NOTE 2: No further sending/receiving of DRB data before the HO is done.
    //   NOTE 3: For AM DRBs the PDCP count is maintained, for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    if (p_TargetPCellId == p_SourceSCellId) {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange, cs_TimingInfo_NR(v_Timing1));
    } else {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s201249 sic@
    }
    
   //At time T2, steps 7-12:
    //7. Target PCell: Inform the SS about the HO and about the source PCell id.
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo_NR(v_Timing2)));
   
    //8. Target PCell: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE 4: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI
    f_NR_CellInfo_SetRNTI(p_TargetPCellId, p_RNTI_Value);
    //Calculate T2 in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), p_TargetPCellId);
    if (isvalue(p_TargetSCellId)) { //@sic R5s210079 sic@
        //@sic R5s201249 sic@
        f_NR_CellInfo_SetRNTI(valueof(p_TargetSCellId), p_RNTI_Value);
        f_NR_SS_C_RNTI_Config(valueof(p_TargetSCellId), p_RNTI_Value, cs_TimingInfo_NR(v_Timing2)); //@sic R5s210389 sic@
    }
    f_NR_SS_C_RNTI_Config(p_TargetPCellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing2), tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369, R5s210389 sic@
    f_NR_SS_RachProcedureConfigHO(p_TargetPCellId, p_RachProcedureConfig, v_TimingInfo, p_RACH_ConfigDedicated);
    
    //9. Target PCell: Activate security.
    if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s190795 R5s200159 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_TargetPCellId, p_NCC, v_TimingInfo);
    } else {
      f_NR_SS_AS_ActivateSecurity_NoRefresh(p_TargetPCellId, v_TimingInfo);  //@sic R5s200102 sic@
    }

    //10. Target PCell: Configure UL grant configuration ("OnSR", periodic TA is not started).
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, v_TimingInfo, -, -, p_UL_GrantConfig_TargetCell);

    //11. Target PCell: Configure Target PCell as PCell.
    //12. Target SCell: Configure Target SCell as SCell with new PCell association.
    //NOTE 5: The FollowOnFlag is set to true in the ASP reconfiguring PCell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
      f_NR_ConfigPCell_AddRelSCells(p_TargetPCellId, {valueof(p_TargetSCellId)}, tsc_PCellIndex_0, v_TimingInfo, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369 sic@
      //Calculate T2 in the TargetSCell
      v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), valueof(p_TargetSCellId));
      f_NR_ConfigSCell(valueof(p_TargetSCellId), p_SCGConfig_SCell, v_TimingInfo);
    }
    
   // After time T2 (without activation time):
    // 13. Target PCell: Receive RRCReconfigurationComplete.
    // when CA is not active , make sure there is no MAC Bearer Routing ( refer step 7-8 of the test case 8.1.4.1.7.1 )
    if(not isvalue(p_TargetSCellId)) //WA#WI=944981
    {
      SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete, -,- )//@sic R5-213516 sic@
    }
    else
    {
    SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete, -, crs_MacBearerRouting(p_TargetPCellId))); //@sic R5-213516 sic@
    }
    //14. Target Pell: Start periodic TA.
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);
    
    //15. Target Pell: Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetPCellId));

    //16. Target PCell: Re-configure RACH procedure as for initial access.
    f_NR_SS_RachProcedureConfig_Def(p_TargetPCellId);
        
    //17. Source PCell: If source PCell is not target SCell
    //17.1 Reset SRBs and release DRBs.
    //17.2 Configure from PCell to normal cell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
        if (p_SourcePCellId != valueof(p_TargetSCellId)) {
            v_DrbIdList := f_NR_GetActiveDRBs(p_SourcePCellId);
            if(ScellBecomingPcell) //WA#WI=944981
            {
                //RB Release and Reset is done after PDCP HO is finished ; other Cell configurations are taken care above in Step 4
                f_NR_SS_SRBs_DRBs_Release(p_SourcePCellId, cs_TimingInfo_Now, v_DrbIdList);  //WA#WI=944981
                f_NR_SS_SRBs_DRBs_Config(p_SourcePCellId, cs_TimingInfo_Now, omit);  //WA#WI=944981
            } //WA#WI=944981
            else
            {
                f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, -, v_DrbIdList,ScellBecomingPcell ); //done above
            } //WA#WI=944981
        }
    }
    
    //18. Source SCell: If source SCell is neither target PCell nor target SCell:
    //18.1 Re-configure source SCell to normal cell and remove source SCell from the SCell list in the source PCell.
    if (not(match(p_SourceSCellId, p_TargetPCellId)) and not match(p_SourceSCellId, p_TargetSCellId)) {
        //f_NR_ReconfigSCellToNormalCell(p_SourceSCellId, v_RadioBearerList, -); //WA#WI=944981
        f_NR_Rel_SCell_Common( p_SourcePCellId,  p_SourceSCellId, tsc_PCellIndex_0, cs_TimingInfo_Now);//Release SCell from PCell @sic R5s201249 sic@ //WA#WI=944981
    }   
  }


0.2 [bookmark: _Toc101446254]Correction to the function f_NR_ReconfigurePCellToNormalCell
	Function name
	f_NR_ReconfigurePCellToNormalCell

	Reason for change
	Function f_NR_ReconfigurePCellToNormalCell is required to have an extra paramter ScellBecomingCell . this boolean flag is to make sure  only Pcell  Normal Cell configuration changes are done where source Scell is becoming Target Pcell. As RB release and reset are required to be done, once the Pdcp Handover is completed

	Summary of change
	Boolean flag ScellBecomingCell is set to True when the function f_NR_ReconfigurePCellToNormalCell is called before Scell becoming Pcell . this flag takes care of doing, only Pcell configurations. 
RB release and Reset for the source Pcell is performed once PDCP HO is completed

	TTCN module
	Common\NR\NR_CommonFunctions_CA.ttcn

	MCC160 Comment
	



    
Before Change:
	  function f_NR_ReconfigurePCellToNormalCell( NR_CellId_Type                          p_PCell,
                                               NR_ServingCellIndex_Type                p_PServingCellIndex,
                                               template (value) TimingInfo_Type        p_TimingInfo := cs_TimingInfo_Now,
                                               IntegerList_Type                        p_DrbToReleassList
                                             ) runs on NR_BASE_PTC
  {
    f_NR_ConfigPCell_AddRelSCells( p_PCell,
                                   {}, //p_SCellList
                                   p_PServingCellIndex,
                                   p_TimingInfo,
                                   omit, // p_RNTI_Value > no need to reconfigure C-RNTI
                                   omit  // p_SS_RadioBearerList > DRB + SRBs to be released just after
                                  );
    //Release DRBs
    f_NR_SS_SRBs_DRBs_Release(p_PCell, p_TimingInfo, p_DrbToReleassList);  //@sic R5s201249 sic@
    //Release and re-establish SRBs - if not ENDC
    if (not f_GetTestcaseAttrib_ENDC(testcasename())) {
        f_NR_SS_SRBs_DRBs_Config(p_PCell, p_TimingInfo, omit);   //@sic R5s201249 sic@
    }
  }



After Change:
	function f_NR_ReconfigurePCellToNormalCell( NR_CellId_Type                          p_PCell,
                                               NR_ServingCellIndex_Type                p_PServingCellIndex,
                                               template (value) TimingInfo_Type        p_TimingInfo := cs_TimingInfo_Now,
                                               IntegerList_Type                        p_DrbToReleassList,
                                               boolean ScellBecomingPcell := false //WA#WI=944981
                                             ) runs on NR_BASE_PTC
  {
    f_NR_ConfigPCell_AddRelSCells( p_PCell,
                                   {}, //p_SCellList
                                   p_PServingCellIndex,
                                   p_TimingInfo,
                                   omit, // p_RNTI_Value > no need to reconfigure C-RNTI
                                   omit  // p_SS_RadioBearerList > DRB + SRBs to be released just after
                                  );
    if(not ScellBecomingPcell)//WA#WI=944981
    {
    //Release DRBs
    f_NR_SS_SRBs_DRBs_Release(p_PCell, p_TimingInfo, p_DrbToReleassList);  //@sic R5s201249 sic@
    //Release and re-establish SRBs - if not ENDC
    if (not f_GetTestcaseAttrib_ENDC(testcasename())) {
        f_NR_SS_SRBs_DRBs_Config(p_PCell, p_TimingInfo, omit);   //@sic R5s201249 sic@
        } //WA#WI=944981
    }
  }



