
	Function name
	f_MCX_InviteMT_SendINVITE

	Reason for change
	As in coversheet

	Summary of change
	As in coversheet

	TTCN module
	MCX_SIP_CallControl


f_MCX_InviteMT_SendINVITE
	  function f_MCX_InviteMT_SendINVITE(MCX_ServiceType_Type p_ServiceType,
                                     MCX_CallType_Type p_CallType := ANY,
                                     MCX_PriorityCallType_Type p_PriorityCallType := NORMAL_CALL,
                                     MCX_CommencementMode_Type p_CommencementMode := NONE,
                                     template (omit) Contact p_Contact := omit,
                                     template (omit) Replaces p_Replaces := omit,
                                     template (omit) SDP_Message p_SDP_Message := omit,
                                     template (value) MCX_MIME_XmlBodies_Type p_XmlBodies := cs_MCX_MIME_XmlBodies_AllOmit,
                                     boolean p_IsReInvite := false)
    runs on MCX_IMS_PTC return IMS_InviteRequestWithSdp_Type
  { /* p_Contact: if omit then the default Contact header according to 36.579-1 Table 5.5.2.5.2-1 is used */
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader_Invite;
    var template (omit) MessageBody v_MessageBody := omit;
    var template (omit) ContentType v_ContentType := omit;
    var IMS_TestProcedure_Type v_TestProcedure;
    var SipUrl v_RequestUrl;

    select (p_ServiceType) {
      case (MCPTT)   { v_TestProcedure := IPCAN_MT_SpeechCall; } 
      case (MCVideo) { v_TestProcedure := IPCAN_MT_VideoCall; }
    }

    f_MCX_CallSetup_ConfigSRTP(p_CallType, v_TestProcedure);
    f_MCX_CallSetup_ConfigSRTP(p_CallType, IPCAN_MO_SpeechCall);  /* @sic R5s211469: SRTP config independent from p_IsReInvite sic@ */

    if (not p_IsReInvite) {
      v_RequestUrl := f_IMS_PTC_ImsInfo_GetContactUrl();         /* SIP URI of the UE's contact address as provided in the Contact-header of the REGISTER message */
      f_MCX_IPCAN_StartSignalling(IPCAN_MT_SpeechCall);
      f_MCX_IPCAN_StartSignalling(v_TestProcedure);
    } else {
      v_RequestUrl := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);   /* SIP URI of the UE's contact address as provided in the Contact-header within any response or request within the dialog */
    }

    if (isvalue(p_XmlBodies.McpttInfo) or
        isvalue(p_XmlBodies.McvideoInfo) or
        isvalue(p_XmlBodies.McdataInfo) or
        isvalue(p_XmlBodies.McpttLocationInfo) or
        isvalue(p_XmlBodies.McvideoLocationInfo) or
        isvalue(p_XmlBodies.McdataLocationInfo)) {
      v_MessageBody := f_MCX_MIME_MessageBodyTX(p_SDP_Message, p_XmlBodies.McpttInfo, p_XmlBodies.McvideoInfo, p_XmlBodies.McdataInfo, p_XmlBodies.McpttLocationInfo, p_XmlBodies.McvideoLocationInfo, p_XmlBodies.McdataLocationInfo);
      v_ContentType := cs_ContentTypeMIME(tsc_MCX_MIME_Boundary);
    } else if (isvalue(p_SDP_Message)) {
      v_MessageBody := cs_MessageBody_SDP(valueof(p_SDP_Message));
      v_ContentType := cs_ContentTypeSDP;
    }

    v_MessageHeader_Invite := f_MCX_InviteRequest_MessageHeaderTX(p_ServiceType, p_PriorityCallType, p_CommencementMode, p_Contact, v_ContentType, p_IsReInvite);
    v_MessageHeader_Invite.replaces := p_Replaces;
    v_InviteRequest := cs_INVITE_Request(v_RequestUrl, v_MessageHeader_Invite, v_MessageBody);
    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, p_SDP_Message);
    IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequest));

    return v_InviteRequestWithSdp;
  }




