
	  function f_IMS_ThreeWaySession_Creation(IMS_TestProcedure_Type p_TestProcedure) runs on IMS_PTC
  {
    var charstring v_ConferenceFactoryUri   := f_IMS_BuildConferenceUri(conferenceFactoryUri);             // @sic R5-150745 sic@
    var charstring v_TemporaryConferenceUri := f_IMS_BuildConferenceUri(temporaryConferenceUri);           // @sic R5-150745 sic@
    var charstring v_FinalConferenceUri     := f_IMS_BuildConferenceUri(finalConferenceUri);               // @sic R5-150745 sic@
    
    var CharStringList_Type v_MediaTypeList := f_IMS_ConferenceEventPackage_MediaTypeList(p_TestProcedure);  // @sic R5-214877: different media for speech call (audio) and video call (speech, video) sic@

    var InviteUserToConference_StateInfo_Type v_Dialog1_StateInfo := valueof(cs_InviteUserToConference_StateInfo_Init(px_IMS_CalleeUri,  px_IMS_CalleeContactUri));
    var InviteUserToConference_StateInfo_Type v_Dialog2_StateInfo := valueof(cs_InviteUserToConference_StateInfo_Init(px_IMS_CalleeUri2, px_IMS_CalleeContactUri2));
    var InviteUserToConference_StateInfo_Type v_Dialog2_StateInfo := valueof(cs_InviteUserToConference_StateInfo_Init(px_IMS_CalleeUri2, px_IMS_CalleeContactUri));

    var boolean v_RadioBearersEstablished := false;

    var template (omit) SUBSCRIBE_Request v_SubscribeRequest;      // optional SUBSCRIBE for conference event package

    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog1;
    var ThreeWaySession_SessionInfo_Type v_SessionInfo_Dialog2;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp_Dialog3;
    var SipUrl v_ContactUrl_UE;

    //--------------------------------------------
    // DIALOG 1
    //--------------------------------------------
    // Setup MO speech call
    f_IMS_CC_StartCall(p_TestProcedure, v_Dialog1_StateInfo.CalleeUri);                                     // C.21/25 Step 1
    
    v_SessionInfo_Dialog1 := fl_IMS_ThreeWaySession_CreateSession(p_TestProcedure, v_Dialog1_StateInfo.CalleeUri, v_Dialog1_StateInfo.CalleeContactUri, v_RadioBearersEstablished);
    v_RadioBearersEstablished := true;
    
    f_IMS_TestBody_Set(true);
    
    //Steps 1 to 4 - Messages in Annex C.8 - Set ongoing call on hold
    // @siclog "Step 1 - 4" siclog@
    f_UT_CallHold(MMI);
    f_IMS_MOCallHoldResume(HOLD, v_SessionInfo_Dialog1.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog1.MessageHeader_PrevResponse, v_SessionInfo_Dialog1.LatestSdpOffer, v_Dialog1_StateInfo.CalleeContactUri);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
 
    //--------------------------------------------
    // DIALOG 2
    //--------------------------------------------
    
    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);
    f_IMS_PTC_ImsInfo_DialogSetSessIdTX("11111112");      // @sic R5-220212 sic@
    
    // @siclog "Step 5" siclog@    Step 1 of C.21
    select (p_TestProcedure) {                            // @sic R5s180053: AT command trigger added to start second call after holding 1st call sic@
      case (IPCAN_MO_SpeechCall) {
        f_UT_RequestIMSCall(MMI, v_Dialog2_StateInfo.CalleeUri);
      }
      case (IPCAN_MO_VideoCall) {
        f_UT_RequestIMSVideoCall(MMI, v_Dialog2_StateInfo.CalleeUri);
      }
    }

    // @siclog "Step 6 - 17" siclog@    Steps defined in Annex C.21 - MO call
    v_SessionInfo_Dialog2 := fl_IMS_ThreeWaySession_CreateSession(p_TestProcedure, v_Dialog2_StateInfo.CalleeUri, v_Dialog2_StateInfo.CalleeContactUri, v_RadioBearersEstablished);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
    
    // @siclog "Step 17A" siclog@
    f_UT_CallMPTY(MMI);  // There is no IPCAN signalling involved => f_IMS_CC_StartCall shall not be used
    // @siclog "Step 17B - 17E" siclog@
    f_IMS_MOCallHoldResume(HOLD, v_SessionInfo_Dialog2.InviteRequestWithSdp.Invite, v_SessionInfo_Dialog2.MessageHeader_PrevResponse, v_SessionInfo_Dialog2.LatestSdpOffer, v_Dialog2_StateInfo.CalleeContactUri);
    
    //--------------------------------------------
    // DIALOG 3
    //--------------------------------------------
    
    f_IMS_PTC_ImsInfo_ActivateDialog(thirdDialog);
    f_IMS_PTC_ImsInfo_DialogSetSessIdTX("11111113");      // @sic R5-220212 sic@
    
    // @siclog "Step 18 - 30" siclog@   Steps defined in Annex C.10
    v_InviteRequestWithSdp_Dialog3 := f_IMS_ConferenceCreation_AnnexC10C38_Steps2_9(p_TestProcedure, v_ConferenceFactoryUri, v_TemporaryConferenceUri, v_FinalConferenceUri, v_RadioBearersEstablished);
    v_ContactUrl_UE := f_MessageHeader_GetContactSipUrl(v_InviteRequestWithSdp_Dialog3.Invite.msgHeader);
    
    //--------------------------------------------
    // Subscribe to conference event package (A.19 steps 10 - 13), invite user A/B (A.20), terminate dialog 1/2
    //--------------------------------------------
    
    // @siclog "Step 27 - 46D" siclog@
    v_SubscribeRequest := f_IMS_ThreeWaySession_SubscribeInvite(v_SessionInfo_Dialog1, v_SessionInfo_Dialog2, v_Dialog1_StateInfo, v_Dialog2_StateInfo, v_FinalConferenceUri, v_MediaTypeList);  /* @sic R5-213528: f_IMS_ThreeWaySession_SubscribeInvite sic@
                                                                                                                                                                                                    @sic R5-214877: v_MediaTypeList sic@ */
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");
    
    //--------------------------------------------
    // Release session
    //--------------------------------------------
    
    // @siclog "Step 47 - 48" siclog@   Terminate conference
    f_IMS_CallReleaseMT(v_ContactUrl_UE);

    //--------------------------------------------
    // Unsubscribe
    //--------------------------------------------

    // @siclog "Step 49 - 50" siclog@
    f_IMS_ConferenceEventPackage_Unsubscribe(v_SubscribeRequest, v_FinalConferenceUri);           /* @sic R5-176937 - new steps sic@ */

    f_IMS_TestBody_Set(false);
  }



