
1. [bookmark: _Toc122434485][bookmark: _Toc95813261]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 8.1.4.4.1 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc95813262]Verification Test Summary 
Test Case: 	8.1.4.4.1
ATS Version:	iwd-TTCN3-B2020-09_D12wk49	
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	Samsung SM-G996B
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc95813263]Corrections required
In addition to the below changes, the common changes of R5s220271 are also required and applied.
2.1 Change 1
	Function name
	fl_TC_8_1_4_4_1_TestBody ()

	Reason for change
	1. At step 2, 10 & 17, The SS transmits an RRCReconfiguration message including ConditionalReconfiguration. Security Keys are not updated as per the specific message contents. Hence there is no need for SS to regenerate the Security Keys. However corresponding parameter of function f_NR5GC_CHO_RRCReconfiguration_SSConfig is set as true by default to indicate Key has been refreshed.
2. RACH preamble is expected at step #3 ,#11, & step 18. However in order for the TTCN to handle this, RACH preamble reception needs to be enabled before receiving and disabled after receiving the RACH.
3. At Step 9: The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 1 and NR Cell 4 as target candidate cells and configure event A5 as trigger event in NR cell 2.  Measurement Object ID shall be added for NR Cell 2 rather than NR Cell 1. p_SourceCellId & p_TargetCellId shall be NR Cell 2 & NR Cell 4 rather than NR Cell 1 & 2.
4. At Step 16: The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration for intra-frequency event A3 and A5 in NR cell 4.  Measurement Object ID shall be added for NR Cell 4 rather than NR Cell 1

	Summary of change
	1. Modified parameter p_RBConfig_KeyChange in function call f_NR5GC_CHO_RRCReconfiguration_SSConfig() to false inorder not to regenerate the security Keys. ( Step 2,  10 & 17)

2. Function calls are applied to enable and disable reception of RACH around steps 3, 11 & 18

3. Step 9: Measurement Object ID is corrected to use NR Cell 2 rather than NR Cell 1. p_SourceCellId & p_TargetCellId is corrected to use NR Cell 2 & NR Cell 4 rather than NR Cell 1 & 2.


4. Step 16: Measurement Object ID is corrected to use NR Cell 4 rather than NR Cell 1.

	TTCN module
	RRC_Conditional_Handover_NR5GC.ttcn

	MCC160 Comment
	



Before Change
	
function fl_TC_8_1_4_4_1_TestBody() runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    //var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;
    var RSRP_Range v_Rsrp;
    var RSRP_Range v_RsrpThres1;
    var RSRP_Range v_RsrpThres2;
    var integer v_DRBId;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasConfig v_MeasConfig_Step_1;
    var template (value) MeasConfig v_MeasConfig_Step_9;
    var template (value) MeasConfig v_MeasConfig_Step_16;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_1;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_9;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_16;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    timer t_Watchdog := 10.0;

    //Power levels of NR cell 1, NR cell 2 and NR cell 4 FR1 acc. Table 8.1.4.4.1.3.2-1, Table 8.1.4.4.1.3.2-2 FFS
    /*         |   Cell 1   |  Cell 2  |  Cell 4
     * -----------------------------------------
     *   T0    |    -82     |   -91    |   -91
     * -----------------------------------------
     *   T1    |    -91     |   -79    |   -91
     * -----------------------------------------
     *   T2    |    -91     |   -91    |   -79
     * -----------------------------------------
     *   T3    |    -79     |   -91    |   -91
     */

   // v_CellPower_Serving_FR1 := -82;
    v_CellPower_Leaving_FR1 := -91;
    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1)){
       v_Rsrp := 2;
       v_RsrpThres1 := 76;
       v_RsrpThres2 := 70;
    }
    else{
       v_Rsrp := 2;          //FFS dummy value FR2
       v_RsrpThres1 := 76;   //FFS dummy value FR2
       v_RsrpThres2 := 70;   //FFS dummy value FR2
    }

    //@siclog "Steps 1-2" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 and NR Cell 4 as target candidate cells and configure event A3 as trigger event in NR cell 1.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 1.
    v_ReportConfigList    := {cs_NR_ReportConfigToAddMod_NR(tsc_NR_IdReportConfigId1, cs_38508_CondTriggerConfig(v_Rsrp)),
                              cs_NR_ReportConfigToAddMod_NR(tsc_NR_IdReportConfigId2, cs_38508_CondTriggerConfig_EventA5(v_RsrpThres1, v_RsrpThres2))};
    v_MeasConfig_Step_1   :=  f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_Cell1)}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf1}, -, -, -, -);
    v_ConditionalReconfiguration_r16_Step_1 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_2EntryCR_1EntryMI(1, tsc_NR_MeasId1, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2),
                                                                                                                                    2, tsc_NR_MeasId1, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell1,
                                             nr_Cell2,
                                             v_DRBInfoList,
                                             -,-,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_1, v_ConditionalReconfiguration_r16_Step_1),
                                             -, -, -); 

    //@siclog "Step 3" siclog@
    //Check: Does the UE initiate a random access procedure transmitting Preamble in NR Cell 2 or in NR cell 4 within 10s?
    t_Watchdog.start;
    alt {
        []SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell2, cr_TimingInfo_Any, ?))
        {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 3, PRACH sent by UE ");
        }
        []t_Watchdog.timeout
        {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
        }
    };

    //@siclog "Step 4" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1"
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);
  

    //@siclog "Step 5" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 2 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell2, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    

    //@siclog "Step 6-7" siclog@
    //The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell1,
                                            nr_Cell2,
                                            -);

    //@siclog "Steps 8" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 2?
    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    f_NR5GC_CheckDataPath(nr_Cell2,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 8");

    //@siclog "Steps 9-10" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 1 and NR Cell 4 as target candidate cells and configure event A5 as trigger event in NR cell 2.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 2.
    v_MeasConfig_Step_9 := f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_Cell1)}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf2});
    v_ConditionalReconfiguration_r16_Step_9 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_2EntryCR_1EntryMI(1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell1),
                                                                                                                                    2, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell1,
                                             nr_Cell2, 
                                             v_DRBInfoList,
                                             -, -,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_9, v_ConditionalReconfiguration_r16_Step_9),
                                             -, -, -, false); //Sujith@81441 NoRefresh

    //@siclog "Step 11" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T2".
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 12" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 4 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell4, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");

    //@siclog "Steps 13-14" siclog@
    //Check:The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 4.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell2,
                                            nr_Cell4,
                                            -);

    //@siclog "Step 15" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 4?
    f_NR5GC_CheckDataPath(nr_Cell4,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 15");

    //@siclog "Steps 16-17" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration for intra-frequency event A3 and A5 in NR cell 4.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 4.
    v_MeasConfig_Step_16 := f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_cell1 )}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf1, cs_NR_MeasId_Config_id2_obj1_conf2});
    v_ConditionalReconfiguration_r16_Step_16 := cs_ConditionalReconfiguration_r16( -,
                                                                                   cs_CondReconfigToAddModList_r16_2EntryCR_2EntryMI(1, tsc_NR_MeasId1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell1),
                                                                                                                                     2, tsc_NR_MeasId1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell4,
                                             nr_Cell1,
                                             v_DRBInfoList,
                                             -, -,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_16, v_ConditionalReconfiguration_r16_Step_16),
                                             -, -, -, false); 

  //@siclog "Step 18" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T3".
    f_NR_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 19" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 1 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");

    //@siclog "Steps 20-21" siclog@
    //The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 1.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell4,
                                            nr_Cell1,
                                            -);

    //@siclog "Step 22" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 1?
    f_NR5GC_CheckDataPath(nr_Cell1,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 22");

    //@siclog "Step 23" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1".
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 24" siclog@
    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 2 within 10s?
    t_Watchdog.start;
    alt {
        []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete))
        {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 24, RRCReconfigurationComplete sent by UE ");
        }
        []t_Watchdog.timeout
        {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");
        }
    };
  }// End of fl_TC_8_1_4_4_1_TestBody




After Change
	
function fl_TC_8_1_4_4_1_TestBody() runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    //var NR_AbsoluteCellPower_Type v_CellPower_Serving_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_EntryCondition_FR1;
    var NR_AbsoluteCellPower_Type v_CellPower_Leaving_FR1;
    var RSRP_Range v_Rsrp;
    var RSRP_Range v_RsrpThres1;
    var RSRP_Range v_RsrpThres2;
    var integer v_DRBId;
    var template (value) ReportConfigToAddModList v_ReportConfigList;
    var template (value) MeasConfig v_MeasConfig_Step_1;
    var template (value) MeasConfig v_MeasConfig_Step_9;
    var template (value) MeasConfig v_MeasConfig_Step_16;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_1;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_9;
    var template (value) ConditionalReconfiguration_r16 v_ConditionalReconfiguration_r16_Step_16;
    var template (value) NR_DRBInfoList_Type v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();
    var PDUSessionInfoList_Type v_PDUSessionInfoList := f_NR5GC_MobileInfo_GetSessionInfoList();
    timer t_Watchdog := 10.0;

    //Power levels of NR cell 1, NR cell 2 and NR cell 4 FR1 acc. Table 8.1.4.4.1.3.2-1, Table 8.1.4.4.1.3.2-2 FFS
    /*         |   Cell 1   |  Cell 2  |  Cell 4
     * -----------------------------------------
     *   T0    |    -82     |   -91    |   -91
     * -----------------------------------------
     *   T1    |    -91     |   -79    |   -91
     * -----------------------------------------
     *   T2    |    -91     |   -91    |   -79
     * -----------------------------------------
     *   T3    |    -79     |   -91    |   -91
     */

   // v_CellPower_Serving_FR1 := -82;
    v_CellPower_Leaving_FR1 := -91;
    v_CellPower_EntryCondition_FR1 := -79;

    v_CellPowerList_AtT1 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
    v_CellPowerList_AtT2 := {cs_NR_CellPower(nr_Cell1, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS
    v_CellPowerList_AtT3 := {cs_NR_CellPower(nr_Cell1, v_CellPower_EntryCondition_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell2, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE),  // FR2 is FFS
                             cs_NR_CellPower(nr_Cell4, v_CellPower_Leaving_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // FR2 is FFS

    if(f_NR_CellInfo_GetIsFR1(nr_Cell1)){
       v_Rsrp := 2;
       v_RsrpThres1 := 76;
       v_RsrpThres2 := 70;
    }
    else{
       v_Rsrp := 2;          //FFS dummy value FR2
       v_RsrpThres1 := 76;   //FFS dummy value FR2
       v_RsrpThres2 := 70;   //FFS dummy value FR2
    }

    //@siclog "Steps 1-2" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 2 and NR Cell 4 as target candidate cells and configure event A3 as trigger event in NR cell 1.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 1.
    v_ReportConfigList    := {cs_NR_ReportConfigToAddMod_NR(tsc_NR_IdReportConfigId1, cs_38508_CondTriggerConfig(v_Rsrp)),
                              cs_NR_ReportConfigToAddMod_NR(tsc_NR_IdReportConfigId2, cs_38508_CondTriggerConfig_EventA5(v_RsrpThres1, v_RsrpThres2))};
    v_MeasConfig_Step_1   :=  f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_Cell1)}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf1}, -, -, -, -);
    v_ConditionalReconfiguration_r16_Step_1 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_2EntryCR_1EntryMI(1, tsc_NR_MeasId1, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2),
                                                                                                                                    2, tsc_NR_MeasId1, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell1,
                                             nr_Cell2,
                                             v_DRBInfoList,
                                             -,-,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_1, v_ConditionalReconfiguration_r16_Step_1),
                                             -, -, -, false); //NoRefresh

    //@siclog "Step 3" siclog@
    //Check: Does the UE initiate a random access procedure transmitting Preamble in NR Cell 2 or in NR cell 4 within 10s?
    t_Watchdog.start;
    alt {
        []SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell2, cr_TimingInfo_Any, ?))
        {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 3, PRACH sent by UE ");
        }
        []t_Watchdog.timeout
        {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");
        }
    };

    //@siclog "Step 4" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1"
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);
    //Reconfigure SS to stop reporting PRACH preambles
    f_NR_SS_SystemIndCtrlConfig(nr_Cell2, cds_NR_SystemIndCtrl_RachPreamble(enable)); 

    //@siclog "Step 5" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 2 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell2, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    //Reconfigure SS to stop reporting PRACH preambles
    f_NR_SS_SystemIndCtrlConfig(nr_Cell2, cds_NR_SystemIndCtrl_RachPreamble(disable)); 

    //@siclog "Step 6-7" siclog@
    //The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell1,
                                            nr_Cell2,
                                            -);

    //@siclog "Steps 8" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 2?
    v_DRBId := f_NR_GetDefaultDRB_ForFirstPDUSession();
    f_NR5GC_CheckDataPath(nr_Cell2,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 8");

    //@siclog "Steps 9-10" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration to set NR Cell 1 and NR Cell 4 as target candidate cells and configure event A5 as trigger event in NR cell 2.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 2.
    v_MeasConfig_Step_9 := f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_Cell2 /*nr_Cell1*/)}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf2});
    v_ConditionalReconfiguration_r16_Step_9 := cs_ConditionalReconfiguration_r16( -,
                                                                                  cs_CondReconfigToAddModList_r16_2EntryCR_1EntryMI(1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell1),
                                                                                                                                    2, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell4)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell2,
                                             nr_Cell4, 
                                             v_DRBInfoList,
                                             -, -,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_9, v_ConditionalReconfiguration_r16_Step_9),
                                             -, -, -, false); //NoRefresh

    f_NR_SS_SystemIndCtrlConfig(nr_Cell4, cds_NR_SystemIndCtrl_RachPreamble(enable)); 
    //@siclog "Step 11" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T2".
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 12" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 4 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell4, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    f_NR_SS_SystemIndCtrlConfig(nr_Cell4, cds_NR_SystemIndCtrl_RachPreamble(disable)); 

    //@siclog "Steps 13-14" siclog@
    //Check:The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 4.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell2,
                                            nr_Cell4,
                                            -);

    //@siclog "Step 15" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 4?
    f_NR5GC_CheckDataPath(nr_Cell4,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 15");

    //@siclog "Steps 16-17" siclog@
    //The SS transmits an RRCReconfiguration message including MeasConfig and ConditionalReconfiguration for intra-frequency event A3 and A5 in NR cell 4.
    //The UE transmits an RRCReconfigurationComplete message in NR cell 4.
    v_MeasConfig_Step_16 := f_NR_GenerateMeasurementConfigNR({cs_MeasObjectId1(nr_Cell4 /*cell1 */)}, v_ReportConfigList, {cs_NR_MeasId_Config_id1_obj1_conf1, cs_NR_MeasId_Config_id2_obj1_conf2});
    v_ConditionalReconfiguration_r16_Step_16 := cs_ConditionalReconfiguration_r16( -,
                                                                                   cs_CondReconfigToAddModList_r16_2EntryCR_2EntryMI(1, tsc_NR_MeasId1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell1),
                                                                                                                                     2, tsc_NR_MeasId1, tsc_NR_MeasId2, f_NR_GenerateCondRRCReconfig_r16_Octetstring(nr_Cell2)));
    f_NR5GC_CHO_RRCReconfiguration_SSConfig( nr_Cell4,
                                             nr_Cell1,
                                             v_DRBInfoList,
                                             -, -,
                                             cs_38508_RRCReconfiguration_CHO(-,v_MeasConfig_Step_16, v_ConditionalReconfiguration_r16_Step_16),
                                             -, -, -, false); //NoRefresh

    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(enable)); 
    //@siclog "Step 18" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T3".
    f_NR_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 19" siclog@
    //Check: Does the UE initiate a random access procedure by transmitting Preamble in NR Cell 1 within 10s?
    t_Watchdog.start;
    SYSIND.receive(car_NR_PRACH_Preamble_IND(nr_Cell1, cr_TimingInfo_Any, ?));
    t_Watchdog.stop;
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");
    f_NR_SS_SystemIndCtrlConfig(nr_Cell1, cds_NR_SystemIndCtrl_RachPreamble(disable)); 

    //@siclog "Steps 20-21" siclog@
    //The SS transmits Random Access Response.
    //The UE transmits RRCReconfigurationComplete message in NR Cell 1.
    f_NR5GC_CHO_RRCReconfigurationComplete( nr_Cell4,
                                            nr_Cell1,
                                            -);

    //@siclog "Step 22" siclog@
    //Check: Does the test result of generic test procedure in TS 38.508-1 Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB1 and NR Cell 1?
    f_NR5GC_CheckDataPath(nr_Cell1,
                          f_NR5GC_GetPdnIndex(v_PDUSessionInfoList[0]),
                          dedicatedBearer,
                          v_DRBId,
                          "Step 22");

    //@siclog "Step 23" siclog@
    //SS adjusts the cell-specific reference signal level according to row "T1".
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 24" siclog@
    //Check: Does the UE transmit RRCReconfigurationComplete message in NR Cell 2 within 10s?
    t_Watchdog.start;
    alt {
        []SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell2, cr_38508_RRCReconfigurationComplete))
        {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 24, RRCReconfigurationComplete sent by UE ");
        }
        []t_Watchdog.timeout
        {
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");
        }
    };
  }// End of fl_TC_8_1_4_4_1_TestBody






1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc95813266]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NR band n78 using NR ciphering nea2 and integrity algorithm nia2
3 [bookmark: _Toc95813267]Execution Log Files
3.1 Samsung SM-G996B
The Samsung SM-G996B UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Keysight/TC_8_1_4_4_1_LOG.html
· PICS/PIXIT parameter file:
Keysight/TC_8_1_4_4_1_PIXIT.xml

