
	Reason for change:
	In the current TTCN implementation, the timer t_Wait is never started.
As a result, the testcase execution stalls at Step 8a1 because the TTCN keeps waiting until guard timer expiry for some UL message.

This needs to be fixed.

	
	

	Summary of change:
	Started the time t_Wait before Step 8a1. 
Please se below.

	Function name
	fl_ReleaseCallWithOptionalDeregister

	Reason for change
	In the current TTCN implementation, the timer t_Wait is never started.
As a result, the testcase execution stalls at Step 8a1 because the TTCN keeps waiting until guard timer expiry for some UL message.

This needs to be fixed.

	Summary of change
	Started the time t_Wait before Step 8a1. 
Please se below.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	



Before Change:
	function fl_ReleaseCallWithOptionalDeregister (NR_CellId_Type p_NR_CellId, float p_WaitTime)  runs on NR5GC_PTC {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var boolean v_RxdMsg := false;
    var default v_MyDefaultVar := null;
    timer t_Wait := p_WaitTime;

    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq();
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(p_NR_CellId));
    
    //@siclog "Step 5-6" siclog@
    //Make the UE release the emergency call.
    f_NR5GC_EmergencyCallRelease(p_NR_CellId);

    deactivate(v_MyDefaultVar);
    if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq()>0) {
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1);
        v_RxdMsg := true;

    } else {
    //@siclog "Step 7" siclog@
    //@siclog "Step 8a1-8b3" siclog@
    //EXCEPTION: Step 8a1 and 8b3 describes optional behaviour that depends on the UE implementation.    
    alt {
      //@siclog "Step 8a1" siclog@
      //The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

      {
        v_RxdMsg := true;
         //@siclog "Step 8a3" siclog@
         //Stop Timer.
         t_Wait.stop;
      }

      //@siclog "Step 8b1" siclog@
      //Timer expires.
      [] t_Wait.timeout {} //@sic R5-216198 sic@
      }
    }
    
    if (v_RxdMsg and f_Check_NG_DeregistrationReqMOMsg (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
     //@siclog "Step 8a2" siclog@
     //The SS transmits a DEREGISTRATION ACCEPT message.
     SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                               cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
    }

}

	




After Change:
	function fl_ReleaseCallWithOptionalDeregister (NR_CellId_Type p_NR_CellId, float p_WaitTime)  runs on NR5GC_PTC {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var boolean v_RxdMsg := false;
    var default v_MyDefaultVar := null;
    timer t_Wait := p_WaitTime;

    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq();
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(p_NR_CellId));
    
    //@siclog "Step 5-6" siclog@
    //Make the UE release the emergency call.
    f_NR5GC_EmergencyCallRelease(p_NR_CellId);

    deactivate(v_MyDefaultVar);
    if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq()>0) {
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1);
        v_RxdMsg := true;

    } else {
    //@siclog "Step 7" siclog@
    //@siclog "Step 8a1-8b3" siclog@
    //EXCEPTION: Step 8a1 and 8b3 describes optional behaviour that depends on the UE implementation.
    t_Wait.start; //WA#11_4_2
    alt {
      //@siclog "Step 8a1" siclog@
      //The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

      {
        v_RxdMsg := true;
         //@siclog "Step 8a3" siclog@
         //Stop Timer.
         t_Wait.stop;
      }

      //@siclog "Step 8b1" siclog@
      //Timer expires.
      [] t_Wait.timeout {} //@sic R5-216198 sic@
      }
    }
    
    if (v_RxdMsg and f_Check_NG_DeregistrationReqMOMsg (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
     //@siclog "Step 8a2" siclog@
     //The SS transmits a DEREGISTRATION ACCEPT message.
     SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                               cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
    }

}





