
1. [bookmark: _Toc122434485][bookmark: _Toc96690746]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 7.24b which is part of the IMS NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
	
	Ramakrishnan Thiruvathirai
	ramakrishnan.thiruvathirai@rohde-schwarz.com

0. [bookmark: _Toc96690747]Verification Test Summary 
Test Case: 	7.24b
ATS Version:	iwd-TTCN3-B2020-09_D12wk49	
System Simulator used:	Keysight S8704A Protocol Conformance Toolset, Rohde & Schwarz 5G Protocol Conformance Test platform
UE used:	Qualcomm SM 8450+SDX65
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc96690748]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc84011980]Change 1
	Function name
	f_TC_7_24b_NR5GC_IMS1()

	Reason for change
	1) The UE will deliver updated SDP Offer within PRACK after 183 Session Progress for the second dialognot currently handled by TTCN.

2) Current TTCN implementation is checking for the wrong callee contact URI for the second dialog.

3) Following agreement of R5-221475 , according to Table 7.24b.4.3-1 sess-id should be “1111111112”. 

4) At steps 30 and 31, according to the conformance requirement on receiving 200 OK the UE will send ACK and then immediately send BYE. These 2 messages can be sent with very little time separation, and if different transport protocols (TCP/UDP) are used then it is possible for the SS to deliver the messages to TTCN in the wrong order (BYE before ACK). 

5) Incorrect test case name is referenced in the dialog2 to-tag.

6) Incorrect step numer in prelinary pass verdict


	Summary of change
	1) SDP offer message check has been added in PRACK receive function
Note : An associated prose CR on 34.229-5 will be raised at RAN5#95

2) px_IMS_CalleeContactUri2 has been passed to the function    f_IMS_MOCallSetup_Send200OK_ReceiveACK() when checking for the 2nd dialog.

3) Updated sess-id as per R5-221475 implementation.

4) Allow TTCN to receive setps 30 and 31  (ACK and BYE) in any order. Parameter is passed to f_IMS_MOCallSetup_Send200OK_ReceiveACK() to not handle ACK inside this function (see change 2)

5) Corrected the to-tag to represent tthe  test case number

6) Correct step number in verdict message.


	TTCN module
	IMS_NR5GC_CallControlTestcases.ttcn

	MCC160 Comment
	



Before Change
	function f_TC_7_24b_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / Forking / UE receives two preliminary responses and one final response / 5GS
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) SDP_Message v_SdpAnswer;
    //var template (present) SDP_Message v_SdpOffer;
    var SipUrl v_ContactUrlUE;
    var SipUrl v_ContactUrlSS;
    var boolean v_AutonomousCallRelease;
    var charstring v_ToTag2 := f_IMS_GenerateTag("TC_7_24a-Dialog-2");

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1 - 9" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1_Step1();
    if (not match(v_InviteRequestWithSdp.Invite.msgHeader.supported, cr_Supported( { "199" } ))) {
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Supported header field does not contain '199' option tag");
    }

    // @siclog "Step 9 - 20" siclog@
    f_IMS_MOCallSetup_NR5GC_A_4_1_Step2_10(v_InviteRequestWithSdp);

    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);
    f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_InviteRequestWithSdp.Invite.msgHeader);  // second dialog
    f_IMS_PTC_ImsInfo_DialogSetLocalTag(v_ToTag2);                                        // second dialog: different To-tag

        f_IMS_PTC_ImsInfo_DialogSetSessIdTX("1111111112",secondDialog);  //R5-221475

    // @siclog "Step 21" siclog@                183 Session progress (dialog 2)
    v_MessageHeader := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequestWithSdp.Invite, px_IMS_CalleeContactUri2);
    v_ContactUrlSS := f_MessageHeader_GetContactSipUrl(v_MessageHeader);
    v_SdpAnswer := f_IMS_BuildSDP_InitialAnswerAudio_TX(v_InviteRequestWithSdp.SdpOffer, -, EVS_16000, 65);
    IMS_Server.send(cas_IMS_DATA_RSP(v_InviteRequestWithSdp.RoutingInfo, cs_Response(c_statusLine183, v_MessageHeader, cs_MessageBody_SDP(v_SdpAnswer))));

    // @siclog "Step 22 - 23" siclog@
    /* v_SdpOffer := f_IMS_BuildSDP_UpdatedOfferAudio_RX(v_InviteRequestWithSdp.SdpOffer, -, EVS_16000);
       f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_MessageHeader, cr_ContentTypeSDP, v_SdpOffer);
       !!!! According to the prose there is no updated SDP offer even though the 2nd remote UE does not the preconditions updated (INVITE: a=curr:qos local/remote none, 183 reponse from the 2nd UE: a=curr:qos local/remote none) !!!! */
    f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_MessageHeader, cr_ContentTypeSDP);

    // @siclog "Step 24 - 26" siclog@
    f_IMS_MOCallSetup_NR5GC_A_4_1_Step8_10(v_InviteRequestWithSdp, px_IMS_CalleeContactUri2);

    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);

    // @siclog "Step 27 - 28" siclog@
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);

    // @siclog "Step 29 - 30" siclog@
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    // @siclog "Step 31 - 32" siclog@
    v_AutonomousCallRelease := true;
    f_IMS_CallReleaseMO(v_InviteRequestWithSdp.Invite, v_ContactUrlSS, -, v_AutonomousCallRelease);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);

    // @siclog "Step 33" siclog@
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    
    f_IMS_TestBody_Set(false);
    
    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);
  }



After Change
	function f_TC_7_24b_NR5GC_IMS1() runs on IMS_PTC
  { // MTSI MO Voice Call / Forking / UE receives two preliminary responses and one final response / 5GS
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) SDP_Message v_SdpAnswer;
    var template (present) SDP_Message v_SdpOffer;
    var SipUrl v_ContactUrlUE;
    var SipUrl v_ContactUrlSS;
    var boolean v_AutonomousCallRelease;
    var charstring v_ToTag2 := f_IMS_GenerateTag("TC_7_24b-Dialog-2");
    timer t_Timer;
    var template (present) MessageHeader v_MessageHeaderACK; 
    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;
    var template (present) MessageHeader v_MessageHeader_Bye;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;
    var template (present) SipUrl v_ContactSipUrl;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);
    
    f_IMS_TestBody_Set(true);

    // @siclog "Step 1 - 9" siclog@
    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);
    v_InviteRequestWithSdp := f_IMS_MOCallSetup_NR5GC_A_4_1_Step1();
    if (not match(v_InviteRequestWithSdp.Invite.msgHeader.supported, cr_Supported( { "199" } ))) {
      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Supported header field does not contain '199' option tag");
    }

    // @siclog "Step 9 - 20" siclog@
    f_IMS_MOCallSetup_NR5GC_A_4_1_Step2_10(v_InviteRequestWithSdp);

    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);
    f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_InviteRequestWithSdp.Invite.msgHeader);  // second dialog
    f_IMS_PTC_ImsInfo_DialogSetLocalTag(v_ToTag2);                                        // second dialog: different To-tag

    // @siclog "Step 21" siclog@                183 Session progress (dialog 2)
    v_MessageHeader := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequestWithSdp.Invite, px_IMS_CalleeContactUri2);
    v_ContactUrlSS := f_MessageHeader_GetContactSipUrl(v_MessageHeader);
    v_SdpAnswer := f_IMS_BuildSDP_InitialAnswerAudio_TX(v_InviteRequestWithSdp.SdpOffer, -, EVS_16000, 65);
    IMS_Server.send(cas_IMS_DATA_RSP(v_InviteRequestWithSdp.RoutingInfo, cs_Response(c_statusLine183, v_MessageHeader, cs_MessageBody_SDP(v_SdpAnswer))));

    // @siclog "Step 22 - 23" siclog@
    /* v_SdpOffer := f_IMS_BuildSDP_UpdatedOfferAudio_RX(v_InviteRequestWithSdp.SdpOffer, -, EVS_16000);
       f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_MessageHeader, cr_ContentTypeSDP, v_SdpOffer);
       !!!! According to the prose there is no updated SDP offer even though the 2nd remote UE does not the preconditions updated (INVITE: a=curr:qos local/remote none, 183 reponse from the 2nd UE: a=curr:qos local/remote none) !!!! */
    v_SdpOffer := f_IMS_BuildSDP_UpdatedOfferAudio_RX(v_InviteRequestWithSdp.SdpOffer, -, EVS_16000);
    f_IMS_MOCallSetup_ReceivePRACK_Send200OK(v_InviteRequestWithSdp, v_MessageHeader, cr_ContentTypeSDP,v_SdpOffer);

    // @siclog "Step 24 - 26" siclog@
    f_IMS_MOCallSetup_NR5GC_A_4_1_Step8_10(v_InviteRequestWithSdp, px_IMS_CalleeContactUri2);

    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);

    // @siclog "Step 27 - 28" siclog@
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    f_IMS_PTC_ImsInfo_ActivateDialog(secondDialog);

    v_MessageHeader_Bye := f_IMS_ByeRequest_MessageHeaderRX(v_InviteRequestWithSdp.Invite); 
    v_ContactSipUrl :=   f_SIP_BuildSipUri_RX(px_IMS_CalleeContactUri2); 
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX( v_InviteRequestWithSdp.Invite, c_statusLine200); /* @sic R5s150692 change 21 sic@ */
    
    // @siclog "Step 29 - 32" siclog@
    f_IMS_MOCallSetup_Send200OK_ReceiveACK(v_InviteRequestWithSdp,px_IMS_CalleeContactUri2, -, -, false);
    //Receive ACK and SIP BYE in any order
    interleave {
      //ACK
      //[] f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_ContactSipUrl, v_MessageHeaderACK)))
      [] IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_ContactSipUrl, v_MessageHeaderACK)))
      {
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 30");
      }  
      //SIP BYE
      [] IMS_Server.receive(car_IMS_Bye_Request(cr_BYE_Request(v_ContactSipUrl, v_MessageHeader_Bye))) -> value v_IMS_DATA_REQ_ByRef
	      {
	        if (f_IMS_PTC_ImsInfo_DialogIsReleased()) {
	          f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Dialog has already been released");
	        }
	        f_IMS_MessageHeader_Request_CheckVia(v_IMS_DATA_REQ_ByRef.Request.Bye.msgHeader, v_IMS_DATA_REQ_ByRef.RoutingInfo.Protocol);    /* @sic R5-134116: additional check of Via header sic@ */
	        f_IMS_PTC_ImsInfo_DialogRelease();    /* there is no dialog anymore; @sic R5s140017 MCC160 implementation sic@ */
	        // --- Send 200 OK
	        v_RoutingInfo := f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ_ByRef.RoutingInfo);
	        IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_IMS_DATA_REQ_ByRef.Request.Bye.msgHeader))));
	      }
    }

    // @siclog "Step 31 - 32" siclog
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 31");

    f_IMS_PTC_ImsInfo_ActivateDialog(firstDialog);

    // @siclog "Step 33" siclog@
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    
    f_IMS_TestBody_Set(false);
    
    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);
  }
	


1. [bookmark: _Toc66864323][bookmark: _Toc84011981]Change 2
	Function name
	f_IMS_MOCallSetup_Send200OK_ReceiveACK()

	Reason for change
	At steps 30 and 31, according to the conformance requirement on receiving 200 OK the UE will send ACK and then immediately send BYE. These 2 messages can be sent with very little time separation, and if different transport protocols (TCP/UDP) are used then it is possible for the SS to deliver the messages to TTCN in the wrong order (BYE before ACK). 


	Summary of change
	Add a new parameter to allow to not handle ACK inside this function


	TTCN module
	IMS_Procedures_CallControl4G5G.ttcn

	MCC160 Comment
	



Before Change
	function f_IMS_MOCallSetup_Send200OK_ReceiveACK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                  charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                  IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                  template (omit) HistoryInfo p_HistoryInfo := omit) runs on IMS_PTC
  { /* @sic R5s130333 change 9.3 - parameter p_ContactUri: acc. to A.2.7 contact in request line of ACK shall be the same in as PRACK sic@ */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5w140112: p_IMS_INVITE_REQ changed to p_InviteRequestWithSdp */
    /* @sic R5s150692 change 21: parameter p_CallType sic@ */
    /* @sic R5-214880: p_HistoryInfo sic@ */
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);
    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;
    var template (value) MessageHeader v_MessageHeader200;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (omit) charstring v_EmergencyTelURI := omit; /* @sic R5s150739 sic@ */
    
    if (p_CallType == ASSERTED_EMERGENCY) { /* @sic R5s150739 sic@ */
      v_EmergencyTelURI := tsc_IMS_Emergency_TelUri;
    }

    // Step 12. Send 200 OK (response to INVITE)
    v_MessageHeader200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, v_EmergencyTelURI, p_ContactUri, -, p_CallType); /* @sic R5s150692 change 21 sic@ @sic R5s150739 sic@ */
    v_MessageHeader200.historyInfo := p_HistoryInfo;
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader200)));
    
    // Step 13. Finally, UE acknowledges call setup.
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, p_CallType); /* @sic R5s150692 change 21 sic@ */
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeaderACK)));
    
    //fl_IMS_Params_CheckSigComp(v_IMS_DATA_REQ.Request.Ack.msgHeader.route.routeBody[0].rrParam);   // acc. to 34.229-1 cl. 13.2.4 step 12 (ACK) "comp=sigcomp" is optional for Route
  }



After Change
	  function f_IMS_MOCallSetup_Send200OK_ReceiveACK(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp,
                                                  charstring p_ContactUri := px_IMS_CalleeContactUri,
                                                  IMS_CallType_Type p_CallType := NORMAL_CALL,
                                                  template (omit) HistoryInfo p_HistoryInfo := omit,
                                                  boolean p_ReceiveAck := true) runs on IMS_PTC 
  { /* @sic R5s130333 change 9.3 - parameter p_ContactUri: acc. to A.2.7 contact in request line of ACK shall be the same in as PRACK sic@ */
    /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */
    /* @sic R5w140112: p_IMS_INVITE_REQ changed to p_InviteRequestWithSdp */
    /* @sic R5s150692 change 21: parameter p_CallType sic@ */
    /* @sic R5-214880: p_HistoryInfo sic@ */
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);
    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;
    var template (value) MessageHeader v_MessageHeader200;
    var template (present) MessageHeader v_MessageHeaderACK;
    var template (omit) charstring v_EmergencyTelURI := omit; /* @sic R5s150739 sic@ */
    
    if (p_CallType == ASSERTED_EMERGENCY) { /* @sic R5s150739 sic@ */
      v_EmergencyTelURI := tsc_IMS_Emergency_TelUri;
    }

    // Step 12. Send 200 OK (response to INVITE)
    v_MessageHeader200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, v_EmergencyTelURI, p_ContactUri, -, p_CallType); /* @sic R5s150692 change 21 sic@ @sic R5s150739 sic@ */
    v_MessageHeader200.historyInfo := p_HistoryInfo;
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader200)));
    
     if(p_ReceiveAck) {
    // Step 13. Finally, UE acknowledges call setup.
    v_MessageHeaderACK := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, p_CallType); /* @sic R5s150692 change 21 sic@ */
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeaderACK)));
    }
    //fl_IMS_Params_CheckSigComp(v_IMS_DATA_REQ.Request.Ack.msgHeader.route.routeBody[0].rrParam);   // acc. to 34.229-1 cl. 13.2.4 step 12 (ACK) "comp=sigcomp" is optional for Route
  }




1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc96690749]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NR band n41 using NR ciphering nea2 and integrity algorithm nia2
3 [bookmark: _Toc96690750]Execution Log Files
3.1 [bookmark: _Toc96690751]Qualcomm SM 8450+SDX65
The Qualcomm SM 8450+SDX65 UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Keysight/TC_7_24b_LOG.html
· PICS/PIXIT parameter file:
Keysight/TC_7_24b_PIXIT.xml

The Qualcomm SM 8450+SDX65 UE passed this test case on Rohde & Schwarz 5G Protocol Conformance Test platform. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_7.24b. LOG.html
· PICS/PIXIT parameter file:
TTCNParameter.rspp

4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc96690752]References 
	[1]
	R5s220374: Supporting information for addition of IMS NR5GC call control test case 7.24b



