
	Reason for change:
	Some issues are introduced in emergency call release sequence following the implementation of R5s220039 and R5s220174.

1. In the current implementation of fl_ReleaseCallWithOptionalDeregister() the timer t_Wait is never started

2. During the  function f_NR5GC_EmergencyCallRelease() the UE may send DEREIGTRATION REQUEST which will be handled in the activated default handler while this function waits for PDU SESSION RELEASE REQUEST (see also change 3). If this has been received and stored, then on expiry of the 5s timer this function should not go on to release the PDU session, because the UE has already locally released so it will not send PDU SESSION RELEASE COMPLETE.

3. The receive statement at the beginning of the altstep a_NR5GC_PDUSessionReleaseReq  will actually match any received UL NAS message. If the message is not PDU SESSION RELEASE REQUEST, the altstep will still execute and the message checking will fail. This means that if a UE sends DEREGISTRATION REQUEST instead of PDU SESSION RELEASE REQUEST. The message will never never get to the default handler which was activated in fl_ReleaseCallWithOptionalDeregister(), and the test case will terminate with the message “PDU Session Release Request failed”


	
	

	Summary of change:
	1. Start timer t_Wait at step 7

2. Add a check to avoid calling f_NR5GC_PDUSessionRelease() if the activated default handler has stored a NAS message (assuming this is DEREGISTRATION REQUEST)

3. Add a template to the SRB.recieve statement so that a UL NAS message other than PDU SESSION RELEASE REQUEST will generate a mismatch, allowing other activated altsteps to be invoked.

	
	

	Consequences if not approved:
	A conformant UE can fail these test cases 

1. [bookmark: _Toc122434485][bookmark: _Toc100327387]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D12wk12 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc100327388]Corrections required

1. [bookmark: _Toc30685521][bookmark: _Toc100327389]Change 1
	Function name
	fl_ReleaseCallWithOptionalDeregister

	Reason for change
	In the current implementation the timer t_Wait is never started

	Summary of change
	Start timer t_Wait at step 7

	TTCN module
	Emergency_NR5GC

	MCC160 Comment
	



Before Change
	function fl_ReleaseCallWithOptionalDeregister (NR_CellId_Type p_NR_CellId, float p_WaitTime)  runs on NR5GC_PTC {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var boolean v_RxdMsg := false;
    var default v_MyDefaultVar := null;
    timer t_Wait := p_WaitTime;

    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq();
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(p_NR_CellId));
    
    //@siclog "Step 5-6" siclog@
    //Make the UE release the emergency call.
    f_NR5GC_EmergencyCallRelease(p_NR_CellId);

    deactivate(v_MyDefaultVar);
    if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq()>0) {
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1);
        v_RxdMsg := true;

    } else {
    //@siclog "Step 7" siclog@
    //@siclog "Step 8a1-8b3" siclog@
    //EXCEPTION: Step 8a1 and 8b3 describes optional behaviour that depends on the UE implementation.
    alt {
      //@siclog "Step 8a1" siclog@
      //The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

      {
        v_RxdMsg := true;
         //@siclog "Step 8a3" siclog@
         //Stop Timer.
         t_Wait.stop;
      }

      //@siclog "Step 8b1" siclog@
      //Timer expires.
      [] t_Wait.timeout {} //@sic R5-216198 sic@
      }
    }
    
    if (v_RxdMsg and f_Check_NG_DeregistrationReqMOMsg (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
     //@siclog "Step 8a2" siclog@
     //The SS transmits a DEREGISTRATION ACCEPT message.
     SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                               cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
    }

}  



After Change
	function fl_ReleaseCallWithOptionalDeregister (NR_CellId_Type p_NR_CellId, float p_WaitTime)  runs on NR5GC_PTC {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var boolean v_RxdMsg := false;
    var default v_MyDefaultVar := null;
    timer t_Wait := p_WaitTime;

    f_NR5GC_MsgInDefault_ResetStoreOfNASMessageReq();
    v_MyDefaultVar := activate(a_NR5GC_AdditionalPDUSessionEstablishmentReq(p_NR_CellId));
    
    //@siclog "Step 5-6" siclog@
    //Make the UE release the emergency call.
    f_NR5GC_EmergencyCallRelease(p_NR_CellId);

    deactivate(v_MyDefaultVar);
    if (f_NR5GC_MsgInDefault_GetNumOfNASMessageReq()>0) {
        v_ReceivedAsp := f_NR5GC_MsgInDefault_GetOneNASMessageReq(1);
        v_RxdMsg := true;

    } else {
    t_Wait.start;
    //@siclog "Step 7" siclog@
    //@siclog "Step 8a1-8b3" siclog@
    //EXCEPTION: Step 8a1 and 8b3 describes optional behaviour that depends on the UE implementation.
    alt {
      //@siclog "Step 8a1" siclog@
      //The UE transmits a DEREGISTRATION REQUEST message with De-registration type IE set to "Normal de-registration".
      [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, ?, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp

      {
        v_RxdMsg := true;
         //@siclog "Step 8a3" siclog@
         //Stop Timer.
         t_Wait.stop;
      }

      //@siclog "Step 8b1" siclog@
      //Timer expires.
      [] t_Wait.timeout {} //@sic R5-216198 sic@
      }
    }
    
    if (v_RxdMsg and f_Check_NG_DeregistrationReqMOMsg (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, crs_DeregisterType('0'B, '0'B,'01'B))) {
     //@siclog "Step 8a2" siclog@
     //The SS transmits a DEREGISTRATION ACCEPT message.
     SRB.send(cas_NR_SRB_NasPdu_REQ(p_NR_CellId, tsc_NR_RbId_SRB2, -,
                               cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));
    }

}  



1. [bookmark: _Toc100327390]Change 2
	Function name
	f_NR5GC_EmergencyCallRelease()

	Reason for change
	During this function, the UE may send DEREIGTRATION REQUEST which will be handled in the activated default handler while this function waits for PDU SESSION RELEASE REQUEST (see also change 3). If this has been received and stored, then on expiry of the 5s timer this function should not go on to release the PDU session, because the UE has already locally released so it will not send PDU SESSION RELEASE COMPLETE.

	Summary of change
	Add a check to avoid calling f_NR5GC_PDUSessionRelease() if the activated default handler has stored a NAS message (assuming this is DEREGISTRATION REQUEST)
Note : The code shown below also incorporates the change of R5s220524

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	



Before Change:
	function f_NR5GC_EmergencyCallRelease(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN); // @sic R5s210363 sic@
    var GSM_MobilityInfo_Type v_PDUSession := valueof(v_GSM_MobilityInfo); // Must be there as previously established it
    var O1_Type v_PDUSessionId := v_PDUSession.SessionId;
    var boolean v_WaitForReq := true;
    timer t_Wait := 5.0;
    
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]); // @sic R5s220039 sic@
    t_Wait.start;
    alt {         // @sic R5-216829 sic@
      [v_WaitForReq] a_NR5GC_PDUSessionReleaseReq (p_NR_CellId, v_PDUSessionId) // @sic R5s211524 sic@
         {
           t_Wait.stop;
           v_WaitForReq := false;
           repeat;
         }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger)  // @sic R5s220039 sic@
         { t_Wait.stop }
      [] t_Wait.timeout   // @sic R5s220039 sic@
         { repeat }
    }

    f_NR5GC_PDUSessionRelease (p_NR_CellId,
                               v_PDUSessionId,
                               cs_GMM_GSM_Cause (omit, '00100100'B));
    f_IP_RtpRtcpLoopback(IP, v_PDUSession.PdnIndex, tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Release");
  }

	




After Change:
	function f_NR5GC_EmergencyCallRelease(NR_CellId_Type p_NR_CellId) runs on NR5GC_PTC
  {
    var template (omit) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), Emergency_PDN); // @sic R5s210363 sic@
    var GSM_MobilityInfo_Type v_PDUSession := valueof(v_GSM_MobilityInfo); // Must be there as previously established it
    var O1_Type v_PDUSessionId := v_PDUSession.SessionId;
    var boolean v_WaitForReq := true; 
    timer t_Wait := 5.0;
    
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS2]); // @sic R5s220039 sic@
    t_Wait.start;
    alt {         // @sic R5-216829 sic@
      [v_WaitForReq] a_NR5GC_PDUSessionReleaseReq (p_NR_CellId, v_PDUSessionId) // @sic R5s211524 sic@
         {
           t_Wait.stop;
           v_WaitForReq := false;
           repeat;
         }
      [] IMS[tsc_Index_IMS2].receive(cmr_IMS_IPCAN_Trigger)  // @sic R5s220039 sic@
         { t_Wait.stop }
      [] t_Wait.timeout   // @sic R5s220039 sic@
         { repeat }
    }

    if(f_NR5GC_MsgInDefault_GetNumOfNASMessageReq()==0) 
    {
        f_NR5GC_PDUSessionRelease (p_NR_CellId,
                               v_PDUSessionId,
                               cs_GMM_GSM_Cause (omit, '00100100'B));
    }
    f_IP_RtpRtcpLoopback(IP, v_PDUSession.PdnIndex, tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Emergency Call Release");
  }



1. [bookmark: _Toc100327391]Change 3
	Function name
	Altstep a_NR5GC_PDUSessionReleaseReq

	Reason for change
	The receive statement at the beginning of the altstep will actually match any received UL NAS message. If the message is not PDU SESSION RELEASE REQUEST, the altstep will still execute and the message checking will fail. This means that if a UE sends DEREGISTRATION REQUEST instead of PDU SESSION RELEASE REQUEST, the message will never never get to the default handler which was activated in fl_ReleaseCallWithOptionalDeregister(), and the test case will terminate with the message “PDU Session Release Request failed”

	Summary of change
	Add a template to the SRB.recieve statement so that a UL NAS message other than PDU SESSION RELEASE REQUEST will generate a mismatch, allowing other activated altsteps to be invoked.

	TTCN module
	NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	



Before Change:
	altstep a_NR5GC_PDUSessionReleaseReq(NR_CellId_Type p_NR_CellId, O1_Type p_PDUSessionId) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
      {
        if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, f_NR5GC_MobileInfo_GetSessionInfoList(), v_ReceivedAsp.Signalling.Nas[0].Pdu, p_PDUSessionId)) {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
        }
      }
  }

	




After Change:
	altstep a_NR5GC_PDUSessionReleaseReq(NR_CellId_Type p_NR_CellId, O1_Type p_PDUSessionId) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo;

    [] SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, {Msg:=?, PiggybackedPduList:={{Msg:=cr_NG_PDU_SESSION_RELEASE_REQUEST(?,?,*,*),PiggybackedPduList:=omit}}}))) -> value v_ReceivedAsp
      {
        if (not f_Check_NG_PDUSessionReleaseReq (v_GSM_MobilityInfo, f_NR5GC_MobileInfo_GetSessionInfoList(), v_ReceivedAsp.Signalling.Nas[0].Pdu, p_PDUSessionId)) {
          f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Release Request failed");
        }
      }
  }





